NLO corrections to the R-ratio

This notebook contains the evaluation of the real and virtual corrections to the R-ration. The virtual diagrams are evaluated using package X, see https://packagex.hepforge.org

T. Becher, Nov. 2020

Auxiliary routines to perform the ϵ-expansion of the phase-space integrals

Distributions

In[889]:=
```
DistApply[ex_, u_] :=
 Collect[ex, {DiracDelta[u], PlusDist[f_, u]}] //.
  distRules[u]
```

In[891]:=
```
distRules[z_] := {a_ DiracDelta[z] :→ (a /. z → 0),
   a_. PlusDist[f_, z] :→ f (a - (a /. z → 0))};
```

In[892]:=
```
nmax = 3;
```

In[893]:=
```
distExp[z_] :=
  z^(-1+a_) → 1/a DiracDelta[z] +
    Sum[a^n/n! PlusDist[Log[z]^n/z, z], {n, 0, nmax}];
```

Example

In[1147]:=
```
exampl = z^(-1+ϵ) (1 - z)^ϵ;
```

Here we can just do the integration

In[1148]:=
```
fullRes = Integrate[exampl, {z, 0, 1},
   Assumptions → Re[ϵ] > 0]
```

Out[1148]= $$\frac{\text{Gamma}[\epsilon]\,\text{Gamma}[1+\epsilon]}{\text{Gamma}[1+2\,\epsilon]}$$

Expand in the integrand using distribution identity:

In[1149]:=
```
exampl2 = exampl /. distExp[z]
```

Out[1149]= $$(1-z)^{\epsilon}\left(\frac{\text{DiracDelta}[z]}{\epsilon}+\text{PlusDist}\left[\frac{1}{z},z\right]+\epsilon\,\text{PlusDist}\left[\frac{\text{Log}[z]}{z},z\right]+\frac{1}{2}\epsilon^2\,\text{PlusDist}\left[\frac{\text{Log}[z]^2}{z},z\right]+\frac{1}{6}\epsilon^3\,\text{PlusDist}\left[\frac{\text{Log}[z]^3}{z},z\right]\right)$$

In[1150]:=
```
exampl3 = Series[exampl2, {ϵ, 0, 2}] // Normal
```

Out[1150]= $$\frac{\text{DiracDelta}[z]}{\epsilon}+\text{DiracDelta}[z]\,\text{Log}[1-z]+\text{PlusDist}\left[\frac{1}{z},z\right]+\epsilon\left(\frac{1}{2}\text{DiracDelta}[z]\,\text{Log}[1-z]^2+\text{Log}[1-z]\,\text{PlusDist}\left[\frac{1}{z},z\right]+\text{PlusDist}\left[\frac{\text{Log}[z]}{z},z\right]\right)+\frac{1}{6}\epsilon^2\left(\text{DiracDelta}[z]\,\text{Log}[1-z]^3+3\,\text{Log}[1-z]^2\,\text{PlusDist}\left[\frac{1}{z},z\right]+6\,\text{Log}[1-z]\,\text{PlusDist}\left[\frac{\text{Log}[z]}{z},z\right]+3\,\text{PlusDist}\left[\frac{\text{Log}[z]^2}{z},z\right]\right)$$

In[1151]:= `exampl4 = DistApply[exampl3, z]`

Out[1151]= $\frac{1}{\epsilon}+\frac{\epsilon\, \text{Log}[1-z]+\frac{1}{2}\epsilon^2\, \text{Log}[1-z]^2}{z}+\frac{\epsilon^2\, \text{Log}[1-z]\, \text{Log}[z]}{z}$

In[1152]:= `example5 = Integrate[exampl4, {z, 0, 1}]`

Out[1152]= $\frac{1}{\epsilon}-\frac{\epsilon\,\pi^2}{6}+2\,\epsilon^2\, \text{Zeta}[3]$

In[1154]:= `Series[fullRes, {ϵ, 0, 2}] // FullSimplify // Normal`

Out[1154]= $\frac{1}{\epsilon}-\frac{\epsilon\,\pi^2}{6}+2\,\epsilon^2\, \text{Zeta}[3]$

Real emission corrections



Parameterization of the 3-particle phase space

This has to be integrated over the three-momenta p_1 and p_2

In[1117]:= `phase =` $\left(\frac{1}{(2\,\pi)^{d-1}}\right)^3 \frac{1}{8\, \text{E1}\, \text{E2}\, \text{E3}}\, \text{DiracDelta}[\text{Q}-\text{E1}-\text{E2}-\text{E3}]\,(2\,\pi)^{d}$;

In[1058]:=
```
solidAngle[d_] = 2 π^(d/2) / Gamma[d/2];
```

Only one nontrivial angular integral: the angle between the three-vectors p_1 and p_2

In[1118]:=
```
phase2 =
  solidAng[d - 1] × solidAng[d - 2] E1^(d-2) E2^(d-2) Sin[θ]^(d-3)
    phase /. E3 → Sqrt[E1^2 + E2^2 + 2 E1 E2 Cos[θ]]
```

Out[1118]=
$$\left(2^{-2 d}\, E1^{-3+d}\, E2^{-3+d}\, \pi^{3-2 d}\, \text{DiracDelta}\left[E1 + E2 - Q + \sqrt{E1^2 + E2^2 + 2\, E1\, E2\, \text{Cos}[\theta]}\right] \text{Sin}[\theta]^{-3+d}\, \text{solidAng}[-2 + d] \times \text{solidAng}[-1 + d]\right) \Big/ \left(\sqrt{E1^2 + E2^2 + 2\, E1\, E2\, \text{Cos}[\theta]}\right)$$

In[1119]:=
```
sol =
  Solve[0 == E1 + E2 - Q + Sqrt[E1^2 + E2^2 + 2 E1 E2 Cos[θ]] /.
      Cos[θ] → Cθ, Cθ][[1]] // Simplify
```

Out[1119]=
$$\left\{C\theta \to \frac{2\, E1\, E2 - 2\, E1\, Q - 2\, E2\, Q + Q^2}{2\, E1\, E2}\right\}$$

In[1120]:=
```
fprime =
  Simplify[
   D[E1 + E2 - Q + Sqrt[E1^2 + E2^2 + 2 E1 E2 Cos[θ]] /.
      Cos[θ] → Cθ, Cθ] /. sol, {Q > E1 + E2}]
```

Out[1120]=
$$-\frac{E1\, E2}{E1 + E2 - Q}$$

Let us rewrite the integral as an integral over the $C\theta = \text{Cos}[\theta]$;

note $d\theta\,\mathrm{Sin}(\theta)^a == d\cos(\theta)\,\mathrm{Sin}(\theta)^{a-1} = d\cos(\theta)\,\left(\mathrm{Sin}^2(\theta)\right)^{\frac{a-1}{2}}$

```
phase3 = FullSimplify[
  1/fprime phase2 /. Sin[θ]^a_ → (1 - Cθ^2)^((a-1)/2) /.
    Cos[θ] → Cθ /. sol, {Q > E1 + E2, E1 > 0, E2 > 0}]
```

Out[1131]= $$4^{-d}\,(E1\,E2)^{-4+d}\,\pi^{3-2d}\left(1-\frac{\left(2\,E1\,E2-2\,(E1+E2)\,Q+Q^2\right)^2}{4\,E1^2\,E2^2}\right)^{\frac{1}{2}(-4+d)}$$
DiracDelta[0] × solidAng[-2 + d] × solidAng[-1 + d]

```
In[1132]:= phase4 =
  2^(4-3 d) π^(3-2 d)
    ((2 E1 + 2 E2 - Q) Q (-2 E1 + Q) (-2 E2 + Q))^(1/2 (-4+d))
    solidAng[-2 + d] × solidAng[-1 + d] // Simplify
```

Out[1132]= $$2^{4-3d}\,\pi^{3-2d}\,\left((2\,E1+2\,E2-Q)\,Q\,(-2\,E1+Q)\,(-2\,E2+Q)\right)^{\frac{1}{2}(-4+d)}$$
solidAng[-2 + d] × solidAng[-1 + d]

```
In[1133]:= FullSimplify[phase4 / phase3 // Together // Factor,
  {Q > E1 + E2, E1 > 0, E2 > 0}]
```

Out[1133]= $$\frac{1}{\mathrm{DiracDelta}[0]}$$

```
In[1134]:= varch = Solve[{y1 == 1 - 2 E1/Q, y2 == 1 - 2 E2/Q}][[1]] // Factor
```

Out[1134]= $$\left\{E1\to-\frac{1}{2}\,Q\,(-1+y1),\ E2\to-\frac{1}{2}\,Q\,(-1+y2)\right\}$$

In[1135]:=
```
jac =
 Det[{D[{E1, E2} /. varch, y1],
   D[{E1, E2} /. varch, y2]}]
```

Out[1135]= $\frac{Q^2}{4}$

In[1136]:=
```
phase5 = FullSimplify[jac phase4 /. varch /. d → 4 - 2 ϵ,
  {Q > 0, y1 > 0, y2 > 0, y1 < 1, y2 < 1}]
```

Out[1136]= $4^{-5+3\epsilon}\,\pi^{-5+4\epsilon}\,Q^2\,\left(-Q^4\,y1\,y2\,(-1+y1+y2)\right)^{-\epsilon}$
$\text{solidAng}[2-2\epsilon]\times\text{solidAng}[3-2\epsilon]$

In[1140]:=
```
phase6 =
 FullSimplify[phase5 /. -1 + y1 + y2 → -y3, {Q > 0}] //
  PowerExpand
```

Out[1140]= $4^{-5+3\epsilon}\,\pi^{-5+4\epsilon}\,Q^{2-4\epsilon}\,y1^{-\epsilon}\,y2^{-\epsilon}\,y3^{-\epsilon}$
$\text{solidAng}[2-2\epsilon]\times\text{solidAng}[3-2\epsilon]$

In[1141]:=
```
phase7 = phase6 /. solidAng[a_] → solidAngle[a] //
  FullSimplify
```

Out[1141]= $\frac{2^{-7+4\epsilon}\,\pi^{-3+2\epsilon}\,Q^{2-4\epsilon}\,y1^{-\epsilon}\,y2^{-\epsilon}\,y3^{-\epsilon}}{\text{Gamma}[2-2\epsilon]}$

In[1142]:= $\text{prefps3} = \frac{Q^2}{128\,\pi^3}\left(\frac{4\,\pi}{Q^2}\right)^{2\epsilon};$

In[1144]:=
```
phase8 = phase7 / prefps3 // PowerExpand // Simplify
```

Out[1144]= $\frac{y1^{-\epsilon}\,y2^{-\epsilon}\,y3^{-\epsilon}}{\text{Gamma}[2-2\epsilon]}$

Amplitude and phase space

3-particle phase space, parameterized in terms of $y_i = 1 - \frac{2E_i}{Q}$.

Note that energy conservation gives $y_1 + y_2 + y_3 = 1$.

The integral goes over the region $0 \le y_1 + y_2 \le 1$.

In[649]:=
$$\text{PS3} = \frac{\text{Q2}}{128\,\pi^3}\left(\frac{4\,\pi}{\text{Q2}}\right)^{2\,\epsilon}\frac{1}{\text{Gamma}[2-2\,\epsilon]}\,(\text{y1 y2 y3})^{-\epsilon};$$

We want to normalize to the tree-level in d dimensions. To do so we also need the two-body phase space **(Exercise)**

In[936]:=
$$\text{PS2} = \frac{2^{-4+4\,\epsilon}\,\pi^{-\frac{1}{2}+\epsilon}\,\text{Q2}^{-\epsilon}}{\text{Gamma}\left[\frac{3}{2}-\epsilon\right]};$$

In[941]:=
$$\text{qqbargAmpSquared} = \text{qqbarAmpSquared}\,\frac{\text{CF}\,\text{g}^2\,\mu 2^{\epsilon}}{\text{Q2}}\,\frac{4\,(\text{y3}-\text{y1 y2}\,\epsilon)+2\,(1-\epsilon)\,\left(\text{y1}^2+\text{y2}^2\right)}{\text{y1 y2}};$$

In[942]:=
$$\text{prefact} = \frac{\alpha\text{s CF}}{4\,\pi}\left(\frac{4\,\pi\,\mu 2}{\text{Q2}}\right)^{\epsilon}\frac{1}{\text{Gamma}[1-\epsilon]};$$

To get the cross section in units of the LO cross section, we compute:

In[1145]:=
$$\text{rest} = \frac{1}{\text{prefact}}\left(\frac{\text{PS3}}{\text{PS2}}\ \text{// FullSimplify // PowerExpand}\right)\frac{\text{qqbargAmpSquared}}{\text{qqbarAmpSquared}}\ \text{/. g} \to \sqrt{4\,\pi\,\alpha\text{s}}\ \text{// PowerExpand}$$

Out[1145]=
$$\text{y1}^{-1-\epsilon}\,\text{y2}^{-1-\epsilon}\,\text{y3}^{-\epsilon}\,\left(2\,(1-\epsilon)\,\left(\text{y1}^2+\text{y2}^2\right)+4\,(-\epsilon\,\text{y1 y2}+\text{y3})\right)$$

... this needs to be integrated over $y_1 + y_2 \le 1$.

In[1146]:=
```
Integrate[y1^(-1-ϵ), {y1, 0, 1}]
```

Out[1146]= $\text{ConditionalExpression}\left[-\frac{1}{\epsilon}, \text{Re}[\epsilon] < 0\right]$

Phase-space integration: distribution expansion

In[1155]:=
```
rest2 = rest /. y3 → 1 - y2 - y1 // PowerExpand
```

Out[1155]= $y1^{-1-\epsilon}\,(1-y1-y2)^{-\epsilon}\,y2^{-1-\epsilon}\left(4\,(1-y1-y2-\epsilon\,y1\,y2)+2\,(1-\epsilon)\left(y1^2+y2^2\right)\right)$

Variable change: $y_2 \to (1-y_1)\,u$. Both variables u and y1 run from 0 ... 1

In[1157]:=
```
rest3 = (1 - y1) rest2 /. y2 → (1 - y1) u /.
   (u (1 - y1))^(-1-ϵ) → u^(-1-ϵ) (1 - y1)^(-1-ϵ)
```

Out[1157]= $u^{-1-\epsilon}\,(1-y1)^{-\epsilon}\,(1-u\,(1-y1)-y1)^{-\epsilon}\,y1^{-1-\epsilon}\left(4\,(1-u\,(1-y1)-y1-\epsilon\,u\,(1-y1)\,y1)+2\,(1-\epsilon)\left(u^2\,(1-y1)^2+y1^2\right)\right)$

Singularities at u = 0 and $y_1 = 0$, but the two integrations are decoupled.

In[1158]:=
```
rest4 =
  Series[rest3 /. distExp[u] /. distExp[y1],
    {ϵ, 0, 0}] // Normal
```

In[1159]:=
```
rest5 = DistApply[DistApply[rest4, u], y1]
```

In[1160]:=
```
rest6 = Integrate[rest5, {y1, 0, 1}]
```

Out[1160]= $\frac{27}{2}+\frac{4}{\epsilon^2}+\frac{7}{\epsilon}-\frac{4\,\pi^2}{3}-u-\frac{2\,u}{\epsilon}+\left(-4+\frac{4}{u}+2\,u\right)\text{Log}[1-u]+2\,(-2+u)\,\text{Log}[u]$

In[1161]:= `rest7 = Integrate[rest6, {u, 0, 1}]`

Out[1161]= $19 + \frac{4}{\epsilon^2} + \frac{6}{\epsilon} - 2\pi^2$

In[1166]:= `rest8 =`

$$\text{Series}\left[\frac{E^{\text{EulerGamma}\,\epsilon}}{\text{Gamma}[1-\epsilon]}\,\text{rest7}, \{\epsilon, 0, 0\}\right] \text{// Normal // Expand}$$

Out[1166]= $19 + \frac{4}{\epsilon^2} + \frac{6}{\epsilon} - \frac{7\pi^2}{3}$

The final result is

In[1167]:= $$\sigma\text{qqbg} = \sigma\text{qqbar} * \frac{\text{CF}\,\alpha\text{s}}{4\pi}\left(\frac{4\pi\, e^{-\text{EulerGamma}}\,\mu^2}{\text{Q2}}\right)^{\epsilon}\left(19 + \frac{4}{\epsilon^2} + \frac{6}{\epsilon} - \frac{7\pi^2}{3}\right);$$

Alternative: subtraction scheme

If $y_i \to 0$ then the E_i= Q/2. This is only possible if the other two particles are flying collinearly in the other direction.

If $y_1 \to 0$ AND $y_2 \to 0$ then $E_3 \to 0$, the gluon becomes soft.

In[1168]:= `rest`

Out[1168]= $\text{y1}^{-1-\epsilon}\,\text{y2}^{-1-\epsilon}\,\text{y3}^{-\epsilon}\left(2\,(1-\epsilon)\,\left(\text{y1}^2+\text{y2}^2\right) + 4\,(-\epsilon\,\text{y1}\,\text{y2}+\text{y3})\right)$

In[1169]:= `collinear13 =`

`Series[rest /. y3 → 1 - y1 - y2, {y2, 0, 0}] // Normal`

Out[1169]= $(1-\text{y1})^{-\epsilon}\,\text{y1}^{-1-\epsilon}\left(-4\,(-1+\text{y1}) - 2\,(-1+\epsilon)\,\text{y1}^2\right)\text{y2}^{-1-\epsilon}$

In[1170]:=
```
collinear23 =
 Series[rest /. y3 → 1 - y1 - y2, {y1, 0, 0}] // Normal
```

Out[1170]= $y1^{-1-\epsilon}\,(1-y2)^{-\epsilon}\,y2^{-1-\epsilon}\left(-4\,(-1+y2)-2\,(-1+\epsilon)\,y2^{2}\right)$

In[1171]:=
```
soft3 = Series[collinear13, {y1, 0, 0}] // Normal
```

Out[1171]= $4\,y1^{-1-\epsilon}\,y2^{-1-\epsilon}$

In[1172]:=
```
soft3alt = Series[collinear23, {y2, 0, 0}] // Normal
```

Out[1172]= $4\,y1^{-1-\epsilon}\,y2^{-1-\epsilon}$

These are the regions in which the cross section becomes singular. Subtracting them, we get a finite cross section which can be integrated numerically:

In[1175]:=
```
finite =
 Series[rest - collinear13 - collinear23 + soft3 /.
   y3 → 1 - y1 - y2, {ϵ, 0, 0}]
```

Out[1175]= $O[\epsilon]^{1}$

.... at higher orders in ϵ there are contributions.

Note: need to *add* the soft part because is contained twice in the collinear results.

Let us integrate the divergent pieces

In[1176]:=
```
collInt13 =
 Integrate[Integrate[collinear13, {y2, 0, 1 - y1},
   Assumptions → Re[ϵ] < 0], {y1, 0, 1},
  Assumptions → Re[ϵ] < 0]
```

Out[1176]= $$\frac{4\,(-4+\epsilon\,(15+(-14+\epsilon)\,\epsilon))\,\text{Gamma}[1-\epsilon]\,\text{Gamma}[-2\,\epsilon]}{\epsilon\,\text{Gamma}[3-3\,\epsilon]}$$

In[1177]:=
```
softInt =
  Integrate[Integrate[soft3, {y2, 0, 1 - y1},
    Assumptions → Re[ϵ] < 0], {y1, 0, 1},
   Assumptions → Re[ϵ] < 0]
```

Out[1177]= $-\frac{4\,\mathrm{Gamma}[1-\epsilon]\,\mathrm{Gamma}[-\epsilon]}{\epsilon\,\mathrm{Gamma}[1-2\epsilon]}$

In[1178]:=
```
sum = Series[2 * collInt13 - softInt, {ϵ, 0, 0}] //
   Normal
```

Out[1178]= $19 + \frac{4}{\epsilon^2} + \frac{6}{\epsilon} - 2\pi^2$

This reproduces the integral from above!

Loop corrections to $e^+ e^- \longrightarrow q\,\bar{q}$

In[1114]:=
```
Needs["X`"]
```

Package X



External line corrections: fermion self energy

Numerator of the diagram

In[1179]:=
```
self = CF g² DiracMatrix[γμ, γ.(p + k), γν]
    (-gμ,ν k.k + ξ kμ kν);
```

$$\frac{1}{(k.k)^2\ (p+k)^2}$$

In[1180]:= `self2 = self // Contract`

Out[1180]= $\text{CF}\ g^2\ \xi\ \text{DiracMatrix}[\gamma.k,\ \gamma.k+\gamma.p,\ \gamma.k] - \text{CF}\ g^2\ \text{DiracMatrix}[\gamma_\nu,\ \gamma.k+\gamma.p,\ \gamma_\nu]\ k.k$

In[1181]:= `self3 = self2 // FermionLineExpand`

Out[1181]= $\text{DiracMatrix}[\gamma.p]\ \left(-2\ \text{CF}\ g^2\ k.k+\text{CF}\ d\ g^2\ k.k-\text{CF}\ g^2\ \xi\ k.k\right) + \text{DiracMatrix}[\gamma.k]\ \left(-2\ \text{CF}\ g^2\ k.k+\text{CF}\ d\ g^2\ k.k+\text{CF}\ g^2\ \xi\ k.k+2\ \text{CF}\ g^2\ \xi\ k.p\right)$

Add denominators and loop integration

In[1182]:= `self4 = LoopIntegrate[self3, k, {k, 0, 2}, {k + p, 0}]`

Out[1182]= $\text{DiracMatrix}[\gamma.p]\ \left(-2\ \text{CF}\ g^2\ \text{PVB}[0,0,p.p,0,0] + \text{CF}\ d\ g^2\ \text{PVB}[0,0,p.p,0,0] - \text{CF}\ g^2\ \xi\ \text{PVB}[0,0,p.p,0,0] - 2\ \text{CF}\ g^2\ \text{PVB}[0,1,p.p,0,0] + \text{CF}\ d\ g^2\ \text{PVB}[0,1,p.p,0,0] - \text{CF}\ g^2\ \xi\ p.p\ \text{PVB}[0,1,p.p,0,0,\text{Weights}\to\{2,1\}]\right)$

In[1183]:= `self5 = LoopRefine[self4]`

Out[1183]= $\text{DiracMatrix}[\gamma.p]\ \left(-\text{CF}\ g^2\ (-1+\xi) - \text{CF}\ g^2\ (-1+\xi)\ \left(\frac{1}{\epsilon} + \text{Log}\left[-\frac{\mu^2}{p.p}\right]\right)\right)$

In[1185]:= `self4onShell = self4 /. p.p → 0`

Out[1185]= $\text{DiracMatrix}[\gamma.p]\ \left(-2\ \text{CF}\ g^2\ \text{PVB}[0,0,0,0,0] + \text{CF}\ d\ g^2\ \text{PVB}[0,0,0,0,0] - \text{CF}\ g^2\ \xi\ \text{PVB}[0,0,0,0,0] - 2\ \text{CF}\ g^2\ \text{PVB}[0,1,0,0,0] + \text{CF}\ d\ g^2\ \text{PVB}[0,1,0,0,0]\right)$

In[1186]:= `self5OnShell = LoopRefine[self4onShell]`

Out[1186]= 0

Note that the package suppresses a factor

In[185]:= $\text{extra} = \frac{i\, E^{-\text{EulerGamma}\,\epsilon}}{(4\,\pi)^{d/2}};$

Vertex correction

In[1187]:= triangle =

$i^6\, CF\, g^2 \langle u[p2, 0], \gamma_\alpha, \gamma.(-p2 + k), \gamma_\mu, \gamma.(p1 + k), \gamma_\beta, u[p1, 0]\rangle (-g_{\alpha,\beta}\, k.k + \xi\, k_\alpha\, k_\beta)$

Out[1187]= $-CF\, g^2 \langle u[p2, 0], \gamma_\alpha, \gamma.k - \gamma.p2, \gamma_\mu, \gamma.k + \gamma.p1, \gamma_\beta, u[p1, 0]\rangle (\xi\, k_\alpha\, k_\beta - k.k\, g_{\alpha,\beta})$

In[1188]:= $\text{triangle2} = \frac{1}{CF\, g^2}\, \text{triangle} \,//\, \text{Contract}$

Out[1188]= $-\xi \langle u[p2, 0], \gamma.k, \gamma.k - \gamma.p2, \gamma_\mu, \gamma.k + \gamma.p1, \gamma.k, u[p1, 0]\rangle + \langle u[p2, 0], \gamma_\beta, \gamma.k - \gamma.p2, \gamma_\mu, \gamma.k + \gamma.p1, \gamma_\beta, u[p1, 0]\rangle\, k.k$

In[1189]:=
```
triangle3 =
  LoopIntegrate[triangle2, k, {k, 0, 2}, {k + p1, 0},
    {k - p2, 0}] /. p1.p1 → 0 /. p2.p2 → 0
```

Out[1189]= $\langle u[p2, 0], \gamma_\mu, u[p1, 0]\rangle$
$(-\xi\, \text{PVA}[0, 0, \text{Weights} \to \{2\}] + 4\, \text{PVB}[0, 0, 0, 0, 0] -$
$6\, \text{PVB}[0, 0, 2\, p1.p2, 0, 0] +$
$d\, \text{PVB}[0, 0, 2\, p1.p2, 0, 0] -$
$4\, p1.p2\, \text{PVC}[0, 0, 0, 0, 2\, p1.p2, 0, 0, 0, 0] +$
$4\, \text{PVC}[1, 0, 0, 0, 2\, p1.p2, 0, 0, 0, 0] -$
$2\, d\, \text{PVC}[1, 0, 0, 0, 2\, p1.p2, 0, 0, 0, 0])$

In[1190]:= `triangle4 = LoopRefine[triangle3]`

Out[1190]= $\langle u[p2, 0], \gamma_\mu, u[p1, 0]\rangle$

$$\left(-8+\frac{\pi^2}{6}-3\left(\frac{1}{\epsilon}+\text{Log}\left[-\frac{\mu^2}{2\,p1.p2}\right]\right)-2\left(\frac{1}{\epsilon^2}+\frac{\text{Log}\left[-\frac{\mu^2}{2\,p1.p2}\right]}{\epsilon}+\frac{1}{2}\text{Log}\left[-\frac{\mu^2}{2\,p1.p2}\right]^2\right)\right)$$

In[1191]:= $\text{triangle5} = \frac{1}{\langle u[p2, 0], \gamma_\mu, u[p1, 0]\rangle}\,\text{triangle4}$ /. p1.p2 → Q2 / 2

Out[1191]=

$$-8+\frac{\pi^2}{6}-3\left(\frac{1}{\epsilon}+\text{Log}\left[-\frac{\mu^2}{Q2}\right]\right)-2\left(\frac{1}{\epsilon^2}+\frac{\text{Log}\left[-\frac{\mu^2}{Q2}\right]}{\epsilon}+\frac{1}{2}\text{Log}\left[-\frac{\mu^2}{Q2}\right]^2\right)$$

Note: $\text{Log}\left[-\frac{\mu^2}{Q2}\right]$ has an imaginary part. The $i\epsilon$'s in the propagators lead to $Q2 \equiv Q2 + i\epsilon$:

Therefore $\text{Log}\left[-\frac{\mu^2}{Q2}\right] = \text{Log}\left[\frac{\mu^2}{Q2}\right] + i\pi$

In[572]:= $\text{triangle6} = \text{triangle5}$ /. $\text{Log}\left[-\frac{\mu^2}{Q2}\right] \to \text{Log}\left[\frac{\mu^2}{Q2}\right] + i\pi$ // Expand // Simplify

In[1193]:=
$$\texttt{triangle7} = \left(\frac{\mu^2}{Q2}\right)^{\epsilon} \left(\texttt{Series}\left[\left(\frac{\mu^2}{Q2}\right)^{-\epsilon} \texttt{triangle6}, \{\epsilon, 0, 0\}\right] \texttt{// Simplify // Normal}\right)$$

Out[1193]=
$$\left(-8 - \frac{2}{\epsilon^2} + \frac{-3 - 2\,i\,\pi}{\epsilon} - 3\,i\,\pi + \frac{7\,\pi^2}{6}\right)\left(\frac{\mu^2}{Q2}\right)^{\epsilon}$$

In[574]:=
$$\texttt{extra} = \frac{i\,E^{-\texttt{EulerGamma}\,\epsilon}}{(4\,\pi)^{d/2}};$$

And the Feynman rules give i * Amp and we divided by g^2 CF

In[579]:=
$$\texttt{prefact} = \frac{1}{i}\,\texttt{extra}\; g^2\, CF \;\texttt{/.}\; g \to \sqrt{4\,\pi\,\alpha s} \;\texttt{/.}\; d \to 4 - 2\,\epsilon \;\texttt{// Simplify}$$

So the full result is

In[581]:=
$$\texttt{triangle8} = \frac{CF\,\alpha s}{4\,\pi}\left(\frac{4\,\pi\,E^{-\texttt{EulerGamma}}\,\mu^2}{Q2}\right)^{\epsilon}\left(-8 - \frac{2}{\epsilon^2} + \frac{-3 - 2\,i\,\pi}{\epsilon} - 3\,i\,\pi + \frac{7\,\pi^2}{6}\right)\langle u[p2, 0], \gamma_{\mu}, u[p1, 0]\rangle;$$

Including the electrons, squaring and writing the result in terms of the tree-level amplitude in d dimensions, we have:

In[593]:=
$$\sigma\texttt{virtual} = \frac{CF\,\alpha s}{4\,\pi}\,\texttt{HoldForm}\left[\left(\frac{4\,\pi\,E^{-\texttt{EulerGamma}}\,\mu^2}{Q2}\right)^{\epsilon}\right]\sigma 0\left(2 * \left(-8 - \frac{2}{\epsilon^2} + \frac{-3}{\epsilon} + \frac{7\,\pi^2}{6}\right) \texttt{// Expand}\right)$$

In[595]:= $\left(19 + \frac{4}{\epsilon^2} + \frac{6}{\epsilon} - \frac{7\,\pi^2}{3}\right) + \left(-16 - \frac{4}{\epsilon^2} - \frac{6}{\epsilon} + \frac{7\,\pi^2}{3}\right)$ // Expand

Checks